MC Integration to find a area/volume



Task: Your task is to find the area of the region that satisfies following two conditions:

(a) $(\sin(2x)\sin y > \frac{1}{4}$

(b) $x^2 + 2y^2 \leq 110$

Algorithm for area (you generalize it for an n-dimensional volume):

1. Enclose the required region in a box of width w and height h
2. Generate 10000 random points in the box that is 10000 pairs of (x_i, y_i)
3. Verify if each point satisfies the condition (a) and (b). Count the points that satisfy the conditions.
4. Area of the blue region = $\frac{\text{number of succeses in step 3}}{10\,000}(w\,h)$
5. Repeat Steps 2 to 4 ten times to obtain 10 values for area of the required region

```
RegionPlot[Sin[2 x] Sin[y] > 1 / 4, {x, -10, 10}, {y, -10, 10}, PlotStyle → Blue]
```




```
RegionPlot[Sin[2 x] Sin[y] > 1 / 4 && x^2 + 2 y^2 < 110,
 {x, -11, 11}, {y, -11, 11}, PlotStyle → Blue]
```



```
RegionPlot[Sin[2 x] Sin[y] > 1 / 4 && x^2 + 2 y^2 < 110, {x, -√110 , √110},
 {y, -Sqrt[110/2], Sqrt[110/2]}, PlotStyle → Blue, AspectRatio → 1/√2]
```



```
MCpoints = 10^4;
trials = 100;
w = 2 √110 ;
h = 2 Sqrt[110/2] ;
AreaOfBox = h w;
For[i = 1, i ≤ trials, i++,
 successes = Count[Table[x = RandomReal[{-w/2, w/2}]; y = RandomReal[{-h/2, h/2}];
     Sin[2 x] Sin[y] > 1 / 4 && x^2 + 2 y^2 < 110, {MCpoints}], True];
 AreaOfBlue[i] = successes/MCpoints AreaOfBox // N;]
measurements = Table[AreaOfBlue[i], {i, 1, trials}]
```

{74.7949, 73.5504, 74.3282, 74.2038, 74.1416, 74.0793, 76.0083, 74.9505, 74.2349, 74.266, 73.9549, 75.7905, 75.7905, 74.9816, 75.1683, 75.1061, 73.3326, 75.915, 72.2748, 74.3905, 73.4571, 72.4304, 75.1061, 74.8572, 73.9549, 71.1547, 74.3282, 74.8572, 72.8659, 71.5281, 74.6083, 76.475, 73.3949, 72.2748, 72.2748, 74.6083, 73.0526, 72.0881, 75.4483, 74.8883, 71.6214, 73.4882, 72.4304, 72.8659, 75.5416, 73.4882, 75.2927, 73.5815, 74.6083, 75.4172, 77.4706, 72.7726, 72.8348, 73.7371, 77.0662, 74.9194, 72.7104, 74.7949, 75.2305, 72.8037, 73.3637, 71.217, 74.7327, 73.5815, 72.3059, 73.7371, 73.5815, 72.4615, 74.0482, 77.5328, 74.2038, 75.6661, 75.5727, 75.3238, 74.3593, 74.2971, 73.2704, 74.3593, 74.8883, 74.7327, 74.0482, 74.7016, 72.8348, 73.1148, 74.7327, 73.4571, 71.6525, 73.5504, 74.0171, 74.6083, 73.1771, 75.5727, 72.0881, 72.617, 73.1771, 72.7104, 72.9904, 72.617, 76.2261, 72.0259}

Mean and Standard Devation when each run was done with 1000 points:

```
Print[Style["⟨area⟩  =  ", 24], Style[Mean[measurements], 24]]
```

⟨area⟩ = 74.21

```
Print[Style["σ  =  ", 24], Style[StandardDeviation[measurements], 24]]
```

σ = 4.16611

Mean and Standard Devation when each run was done with

10000 points:

```
Print[Style["⟨area⟩  =  ", 24], Style[Mean[measurements], 24]]
```

⟨area⟩ = 74.0075

```
Print[Style["σ  =  ", 24], Style[StandardDeviation[measurements], 24]]
```

σ = 1.34996

MC Integration to get total observed cross-section in Bhabha scattering

Task: Your task is to find the total cross-section between $\theta = 0.4$ and $\theta = \pi - 0.4$ by MC integration method for Bhabha scattering cross-section.

$$\sigma\ (\Theta) = \frac{1 + \cos^4 \frac{\Theta}{2}}{\sin^4 \frac{\Theta}{2}} - \frac{2 \cos^4 \frac{\Theta}{2}}{\sin^2 \frac{\Theta}{2}} + \frac{1 + \cos^2 \Theta}{2} \tag{1}$$

Following are the steps to do MC Integration in general:

1. Obtain N_{max} random values of θ in the region of interest $(\theta_{\text{min}}, \theta_{\text{max}})$
2. Evaluate the integrand at all the values of θ from step 1, and store it in an array
3. Calculate the Mean of values in array from step 2. Call this mean μ.
4. integral = $\mu\,(\theta_{\text{max}} - \theta_{\text{min}})$
5. Repeat Steps 1 to 4 hundred times to obtain 100 values for the integral
6. Calculate mean and standard deviation of the integral

For the Bhabha scattering cross-section we will follow stratified sampling method. That is we will divide the entire region in two regions and generate a different number of points in each region according to its approximate standard deviation. First define the cross-section function:

```
diffCrossSection[θ_] = (1 + Cos[θ / 2]^4)/Sin[θ / 2]^4 - (2 Cos[θ / 2]^4)/Sin[θ / 2]^2 + (1 + Cos[θ]^2)/2;
```

Here is the plot of the cross-section function in the range $\theta = 0.4$ radians to $\theta = \pi - 0.4$ radians.

```
Plot[{diffCrossSection[θ]}, {θ, 0.4, π - 0.4}, PlotStyle → Thickness[0.005],
 AxesLabel → {"θ", "dσ/dΩ"}, Frame → True, PlotRange → Full]
```



We note that most of the cross-section comes from region $\theta < 1.0$. So we should pick more points in that region and less for $\theta > 1.0$. Below are the plots in the two regions:

```
Plot[{diffCrossSection[θ]}, {θ, 0.4, 1}, PlotStyle → Thickness[0.005],
 AxesLabel → {"θ", "dσ/dΩ"}, Frame → True, PlotRange → Full]
```



```
Plot[{diffCrossSection[θ]}, {θ, 1, π - 0.4}, PlotStyle → Thickness[0.005],
 AxesLabel → {"θ", "dσ/dΩ"}, Frame → True, PlotRange → Full]
```



How many points we should choose in each range of θ? This is decided by the following relation: (subscript a and b corresponds to two regions and N_{total} is the total number of random numbers available to you or you can generate in alloted computer time)

$$N_a = \frac{\sigma_a}{\sigma_a + \sigma_b} N_{total}, \text{ and } N_b = \frac{\sigma_b}{\sigma_a + \sigma_b} N_{total} \quad (2)$$

So we need to get at least a rough estimate of $\sigma_{a,b}$ to get the idea. This we will do by evaulating cross-section at min, max and mid point in each range and calculating st. dev.

```
σa = diffCrossSection[{0.4, (1.0 + 0.4)/2, 1.0}] // StandardDeviation
```

646.722

```
σb = diffCrossSection[{1.0, (1.0 + π - 0.4)/2, π - 0.4}] // StandardDeviation
```

13.4111

Taking $N_{\text{total}} = 10\,000$, we get

```
Ntotal = 10^4;
Na = σa/(σa + σb) Ntotal // Round
```

9797

```
Nb = σb/(σa + σb) Ntotal // Round
```

203

Now we calculate the area in each region accoring to the algorithm. First get the array of random θ_i in two ranges:

```
θa = RandomReal[{0.4, 1}, Na];
θb = RandomReal[{1, π - 0.4}, Nb];
```

Now evaluate cross-section at these values:

```
CSlista = diffCrossSection[θa];
CSlistb = diffCrossSection[θb];
```

Now calculate their mean and multiply by their respective ranges:

```
CSa = Mean[CSlista] (1 - 0.4);
CSb = Mean[CSlistb] (π - 0.4 - 1);
```

Get total crosssection by adding CS in two regions:

```
CS = CSa + CSb
```

149.061

This was result of one Monte Carlo experiment. I want to repeat this at least 100 times to get a distribution in the cross-section value obtained by this method. Remember, since θ_i are random variables then so is the cross-section obtained by MC method, thus it must have a distrbution. To do the loop I will combine the steps of previous lines in one line here and use the power of *Mathematica*: (I am going to do this for $N_{a,b} \leftarrow 100\, N_{a,b}$ to get a nice histogram and more precise result)

```
CSbyMClist = Table[Mean[diffCrossSection[RandomReal[{0.4, 1}, 100 Na]]] (1 - 0.4) + Mean[
        diffCrossSection[RandomReal[{1, π - 0.4}, 100 Nb]]] (π - 0.4 - 1), {100}]; // Timing
```

{78.0941, Null}

```
Histogram[CSbyMClist, 10]
```



Mean and Standard Deviation for this purpose are:

```
Mean[CSbyMClist]
```

152.076

```
StandardDeviation[CSbyMClist]
```

0.145636

```
TrueValue = Integrate[diffCrossSection[θ], {θ, 0.4, π - 0.4}] // Timing
```

{0.086097, 152.081}